
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 6.2.1.5 which is part of the ATS iwd-TTCN3-B2019-06_D20wk11 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Fu Wang

wangfu@datangmobile.cn
                                Yongsheng Ma
                                      mayongsheng@starpointcomm.com
1.1 Verification Test Summary

Test Case: 
6.2.1.5
ATS Version:
iwd-TTCN3-B2019-06_D20wk11
System Simulator used:
Datang LinkTester 5G Protocol Conformance Test System ECT9610
                                              Starpoint SP9500 Test system
UE used:
HiSilicon Balong 5000
Verification Status:
PASS
2 Corrections required
2.1 Change 1
	Function name
	fl_TC_6_2_1_5_TestBody

	Reason for change
	1. In order to cooperate with the registration process of LTE side in step 3, NR frequency point information should be sent to LTE side.

2. According to TS 24.301 cl 5.5.1.2.4, the interworking without N26 interface indicator is valid in RPLMN and EPLMNs. UE will treat no N26 interface when UE performs mobility registration from S1 mode to N1 mode at step 21, because PLMN1 (00101) of NR Cell 1 is not equivalent for RPLMN (PLMN15 00111), so the exclusive S1 to N1 mode process should be used.

	Summary of change
	1. Add function fl_NR5GC_IRAT_UECapabilityTransfer before step 3.
2. Replace f_NR5GC_IRAT_S1ToN1_IdleMode with f_NR5GC_IRAT_S1ToN1_IdleMode_6215 in step 21.

	TTCN module
	Idle_PLMNSelection_NR5GC_NR.ttcn

	MCC160 Comment
	1. Accepted in principle, but it should be function f_NR5GC_IRAT_UECapabilityTransfer();
2. Accepted in principle, but implemented differently. Reuse function f_NR5GC_IRAT_S1ToN1_IdleMode()


Before change

function fl_TC_6_2_1_5_TestBody() runs on NR5GC_PTC

  {

    var Q_RxLevMin v_Q_RxLevMin;

    var DeltaValues_Type v_DeltaPrimaryBand;

    timer t_Tmin;

    timer t_Tmax;

    // The SS adjusts the NR and E-UTRAN Cell power levels according to row T0

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);

    // @siclog "Step 2" siclog@

    // SS starts timer of tmin = 2 minutes and tmax =(6 minutes + cell selection time)

    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA);

    t_Tmin.start(f_NR_SetTimerToleranceMin(nr_Cell1, nonProtocolTimer, 120.0));

    t_Tmax.start(f_NR_SetTimerToleranceMax(nr_Cell1, nonProtocolTimer, 660.0));

    // @siclog "Step 19" siclog@

    // The SS adjusts cell levels according to row T1 of table 6.2.1.5.3.2-1.

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {

        v_DeltaPrimaryBand := f_NR_MobileInfo_GetPrimaryDeltaValue();

        v_Q_RxLevMin := f_Round((91.0 + int2float(v_DeltaPrimaryBand.DeltaNRf1))/2.0);

        f_NR_CellInfo_SetSIB1_CellSelectionInfo(nr_Cell1, cs_38508_SIB1_CellSelectionInfo(v_Q_RxLevMin));

        f_NR_ModifySysinfo(nr_Cell1, false);

    }

    f_NR_SetCellPower (nr_Cell1, tsc_NR_SuitableInterFreqCellSSS_EPRE_FR1, tsc_NR_SuitableInterFreqCellSSS_EPRE_FR2);

    // @siclog "Step 20" siclog@

    // Check: Does the UE send an RRCSetupRequest on NR Cell 1 after tmin but before tmax expires?

    alt {

      [] SRB.receive( car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any)))

        {

          // verdict fail RRCSetupRequest received on NR Cell 1 within tmin

          f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

        }

      [] t_Tmin.timeout

        {

          alt {

            [] SRB.check(receive( car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))))

             {

               t_Tmax.stop;

               // verdict pass RRCSetupRequest received on NR Cell 1 after t_Tmin time but before tmax

               f_NR_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

             }

            [] t_Tmax.timeout

             {

              // verdict fail No RRCSetupRequest received

              f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

             }

          }

        }

    }

    // @siclog "Step 20-27a1" siclog@

    // Steps 1-7a1 of the registration procedure described in TS 38.508-1[4] table 4.9.9.2.2-1 are performed on NR Cell 1

    f_NR5GC_IRAT_S1ToN1_IdleMode(nr_Cell1);
  }
After change

function fl_TC_6_2_1_5_TestBody() runs on NR5GC_PTC

  {

    var Q_RxLevMin v_Q_RxLevMin;

    var DeltaValues_Type v_DeltaPrimaryBand;

    timer t_Tmin;

    timer t_Tmax;

    // The SS adjusts the NR and E-UTRAN Cell power levels according to row T0

    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);

    fl_NR5GC_IRAT_UECapabilityTransfer(nr_Cell1);
    // @siclog "Step 2" siclog@

    // SS starts timer of tmin = 2 minutes and tmax =(6 minutes + cell selection time)

    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA);
    t_Tmin.start(f_NR_SetTimerToleranceMin(nr_Cell1, nonProtocolTimer, 120.0));

    t_Tmax.start(f_NR_SetTimerToleranceMax(nr_Cell1, nonProtocolTimer, 660.0));

    // @siclog "Step 19" siclog@

    // The SS adjusts cell levels according to row T1 of table 6.2.1.5.3.2-1.

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {

        v_DeltaPrimaryBand := f_NR_MobileInfo_GetPrimaryDeltaValue();

        v_Q_RxLevMin := f_Round((91.0 + int2float(v_DeltaPrimaryBand.DeltaNRf1))/2.0);

        f_NR_CellInfo_SetSIB1_CellSelectionInfo(nr_Cell1, cs_38508_SIB1_CellSelectionInfo(v_Q_RxLevMin));

        f_NR_ModifySysinfo(nr_Cell1, false);

    }

    f_NR_SetCellPower (nr_Cell1, tsc_NR_SuitableInterFreqCellSSS_EPRE_FR1, tsc_NR_SuitableInterFreqCellSSS_EPRE_FR2);

    // @siclog "Step 20" siclog@

    // Check: Does the UE send an RRCSetupRequest on NR Cell 1 after tmin but before tmax expires?

    alt {

      [] SRB.receive( car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any)))

        {

          // verdict fail RRCSetupRequest received on NR Cell 1 within tmin

          f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

        }

      [] t_Tmin.timeout

        {

          alt {

            [] SRB.check(receive( car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))))

             {

               t_Tmax.stop;

               // verdict pass RRCSetupRequest received on NR Cell 1 after t_Tmin time but before tmax

               f_NR_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

             }

            [] t_Tmax.timeout

             {

              // verdict fail No RRCSetupRequest received

              f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.2.1.5 Step 20");

             }

          }

        }

    }

    // @siclog "Step 20-27a1" siclog@

    // Steps 1-7a1 of the registration procedure described in TS 38.508-1[4] table 4.9.9.2.2-1 are performed on NR Cell 1

    f_NR5GC_IRAT_S1ToN1_IdleMode_6215(nr_Cell1);
  }
2.2 Change 2
	Function name
	f_NR5GC_IRAT_S1ToN1_IdleMode_6215

	Reason for change
	1. Because UE will treat no N26 interface for this case. Consider all the UE vendors, It should set the IE PDU session status into ‘INACTIVE’ in the message REGISTRATION ACCEPT in order to trigger all of the UEs to locally release the PDU sessions and perform PDU session establishment procedure after mobility registration. PDU session establishment procedure shall be added in test body.

	Summary of change
	1. Use f_NR5GC_PDUSessionEstablishment to complete the corresponding steps

	TTCN module
	NR5GC_IRAT_CommonFunctions.ttcn

	MCC160 Comment
	1. Accepted in principle, but implemented differently. Reuse function f_NR5GC_IRAT_S1ToN1_IdleMode()


New function
function f_NR5GC_IRAT_S1ToN1_IdleMode_6215(NR_CellId_Type p_CellId,

                                        RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,

                                        boolean p_FirstTimeS1ToN1 := true,

                                        template (omit) NG_MobileIdentity p_GUTIFromPreamble := omit,

                                        template NG_TrackingAreaId p_TAI := *,

                                        template (omit) NG_MobileIdentity p_GUTIToSend := omit,

                                        template (omit) NG_TrackingAreaIdList p_TAIList := omit,

                                        boolean p_Registered_Previously_5GCN := true)  runs on NR5GC_PTC

  {

    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_CellId);

    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (p_CellId);

    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);

    var template (value) NG_MobileIdentity v_GUTIFromPreamble;

    var template NG_TrackingAreaId v_TAI;

    var RRCConnectionReleaseRequired_Type v_ReleaseConnection := p_ReleaseConnection;

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(EUTRA);

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);

    if (p_FirstTimeS1ToN1) { // If this is the 1st time then must release RRC Connection before re-registration and PDU session Modification

      v_ReleaseConnection := noRrcConnectionRelease;

    }

    if (isvalue(p_GUTIFromPreamble)) {

      v_GUTIFromPreamble := valueof(p_GUTIFromPreamble);

    } else {

      v_GUTIFromPreamble := f_NG_GutiParameters2MobileIdentity (v_PLMN, v_GutiParams);

    }

    if (isvalue(p_TAI)) {

      v_TAI := valueof(p_TAI);

    } else {

      v_TAI := f_LastVisited_TAI (p_CellId);

    }

   f_NR5GC_IRAT_MobilityRegistration_6215 (p_CellId,

                                      v_SecurityRx,

                                      f_NR_Convert_CoOrdTo5G_GUTI(v_CoOrd_Guti, omit),

                                      omit, //  @sic R5-198854 sic@

                                      v_GUTIFromPreamble,

                                      v_TAI,

                                      p_GUTIToSend,

                                      p_TAIList,

                                      v_ReleaseConnection,

                                      p_Registered_Previously_5GCN);

    // always wait for PDU info, even if not 1st time

   if (p_FirstTimeS1ToN1) {

      f_NR5GC_PDUSessionEstablishment (p_CellId, pc_noOf_PDUs, tsc_NR_RbId_SRB1);

    }

   if (p_ReleaseConnection == rrcConnectionRelease) {

      // Release RRC to enter RRC_IDLE

      f_NR_RRCRelease(p_CellId);

    }

  }
2.3 Change 3
	Function name
	f_NR5GC_IRAT_MobilityRegistration_6215

	Reason for change
	1. According to TS 33.501 clause 8.2, a complete TAU Request message shall be contained in Registration Request when UE perform mobility from EPS to 5GS over N26. Considering there is not detailed enough specification for the scenario at step 21, which may be decided by UE’s implementation. Thus, the TAU Request message included in EPS NAS message container shall set to ‘not checked’ to keep compatible with different implementations of UE vendors.
2. In order to cooperate with change 2, PDU_SessionStatus in registration request message is not checked, and empty PDU_SessionStatus is carried in registration accept message

	Summary of change
	1. Not check EPS_MessageContainer in f_Check_NG_RegistrationReqMsg.
2. Not check PDU_SessionStatus in f_Check_NG_RegistrationReqMsg, set v_GMM_MobilityInfo.SessionStatus to cs_PDU_SessionStatusNull

	TTCN module
	NR5GC_IRAT_CommonFunctions.ttcn

	MCC160 Comment
	1. Accepted, but implemented differently.
2. Accepted in principle, but implemented differently.
Reuse function f_NR5GC_IRAT_MobilityRegistration() & f_NR5GC_IRAT_MobilityRegistration()


New function
function f_NR5GC_IRAT_MobilityRegistration_6215 (NR_CellId_Type p_CellId,

                                              EUTRA_NR_CoOrd_Security_Type p_EUTRA_Auth_Params,

                                              template (present) NG_MobileIdentity p_MappedGUTI,

                                              template (omit) NAS_KeySetIdentifier p_NonCurrentKSI := omit,

                                              template (omit) NG_MobileIdentity p_GUTIFromPreamble,

                                              template NG_TrackingAreaId p_TAI := *,

                                              template (omit) NG_MobileIdentity p_GUTIToSend := omit,

                                              template (omit) NG_TrackingAreaIdList p_TAIList := omit,

                                              RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,

                                              boolean p_Registered_Previously_5GCN := true)  runs on NR5GC_PTC

  { //@sic R5-198854: added p_Registered_Previously_5GCN sic@

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NasCount_Type v_NasCount;

    var template (present) B1_Type v_Tsc := '0'B;                                 //@sic R5-198854 sic@

    var template (present) NAS_KsiValue v_KeySetId := p_EUTRA_Auth_Params.KSI;    //@sic R5-198854 sic@

    var TypeOfRegistration_Type v_RegType := Initial_Secure;

    if ( not p_Registered_Previously_5GCN) { //@sic R5-198854 sic@

        v_KeySetId := '111'B;

        v_Tsc := '?'B;//('1'B, '0'B);

        v_RegType := Initial_NoSecurity;

    }

    // Steps 1 - 3

    // RRC Connection Setup

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (p_CellId, tsc_NR_RRC_TI_Def, ?, (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected));

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo,

                                           v_ReceivedMsg.Pdu,

                                           Mobility,

                                           p_MappedGUTI,

                                           v_KeySetId,

                                           v_Tsc,

                                           p_NonCurrentKSI,

                                           cr_NG_GMM_Cap(?, ?, '1'B),

                                           -,-,

                                           p_TAI,

                                           cdr_NR5GC_UENetworkCap('17'O),

                                           -,

                                           -,

                                           -,

                                           cr_UEStatus_S1,

                                           p_GUTIFromPreamble)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    // @sic R5-198854 sic@

    //Steps 4-9: 5GMM Authentication and 5GMM security

    v_NasCount :=f_NR5GC_RRC_Idle_Steps5_9_AKA (v_GMM_MobilityInfo, p_CellId, v_ReceivedMsg, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_CellId)), v_RegType);

    f_NR5GC_MobileInfo_SetGMM_MobilityInfo(v_GMM_MobilityInfo);

    //Steps 8-9 RRC security activation

    f_NR_RRC_ActivateSecurity (p_CellId, v_NasCount);

    v_GMM_MobilityInfo.SessionStatus := valueof(cs_PDU_SessionStatusNull);

    f_NR5GC_MobilityRegistration_Steps4_6(p_CellId,v_GMM_MobilityInfo, p_GUTIToSend, p_TAIList, p_ReleaseConnection);

  }
New function
function f_NR5GC_IRAT_MobilityRegistration_6215 (NR_CellId_Type p_CellId,

                                              EUTRA_NR_CoOrd_Security_Type p_EUTRA_Auth_Params,

                                              template (present) NG_MobileIdentity p_MappedGUTI,

                                              template (omit) NAS_KeySetIdentifier p_NonCurrentKSI := omit,

                                              template (omit) NG_MobileIdentity p_GUTIFromPreamble,

                                              template NG_TrackingAreaId p_TAI := *,

                                              template (omit) NG_MobileIdentity p_GUTIToSend := omit,

                                              template (omit) NG_TrackingAreaIdList p_TAIList := omit,

                                              RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,

                                              boolean p_Registered_Previously_5GCN := true)  runs on NR5GC_PTC

  { //@sic R5-198854: added p_Registered_Previously_5GCN sic@

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NasCount_Type v_NasCount;

    var template (present) B1_Type v_Tsc := '0'B;                                 //@sic R5-198854 sic@

    var template (present) NAS_KsiValue v_KeySetId := p_EUTRA_Auth_Params.KSI;    //@sic R5-198854 sic@

    var TypeOfRegistration_Type v_RegType := Initial_Secure;

    if ( not p_Registered_Previously_5GCN) { //@sic R5-198854 sic@

        v_KeySetId := '111'B;

        v_Tsc := '?'B;//('1'B, '0'B);

        v_RegType := Initial_NoSecurity;

    }

    // Steps 1 - 3

    // RRC Connection Setup

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (p_CellId, tsc_NR_RRC_TI_Def, ?, (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected));

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo,

                                           v_ReceivedMsg.Pdu,

                                           Mobility,

                                           p_MappedGUTI,

                                           v_KeySetId,

                                           v_Tsc,

                                           p_NonCurrentKSI,

                                           cr_NG_GMM_Cap(?, ?, '1'B),

                                           -,-,

                                           p_TAI,

                                           cdr_NR5GC_UENetworkCap('17'O),

                                           -,

                                           -,

                                           -,

                                           cr_UEStatus_S1,

                                           p_GUTIFromPreamble)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    // @sic R5-198854 sic@

    //Steps 4-9: 5GMM Authentication and 5GMM security

    v_NasCount :=f_NR5GC_RRC_Idle_Steps5_9_AKA (v_GMM_MobilityInfo, p_CellId, v_ReceivedMsg, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_CellId)), v_RegType);

    f_NR5GC_MobileInfo_SetGMM_MobilityInfo(v_GMM_MobilityInfo);

    //Steps 8-9 RRC security activation

    f_NR_RRC_ActivateSecurity (p_CellId, v_NasCount);

    v_GMM_MobilityInfo.SessionStatus := valueof(cs_PDU_SessionStatusNull);

    f_NR5GC_MobilityRegistration_Steps4_6(p_CellId,v_GMM_MobilityInfo, p_GUTIToSend, p_TAIList, p_ReleaseConnection);

  }
2.4 Change 4
	Function name
	cs_PDU_SessionStatusNull

	Reason for change
	1. Add corresponding template due to change 2.3

	Summary of change
	1. Add new template cs_PDU_SessionStatusNull and set psi0_7 and psi8_15 to '00000000'B

	TTCN module
	NG_NAS_Templates.ttcn

	MCC160 Comment
	Accepted


New template
template (present) PDU_SessionStatus cs_PDU_SessionStatusNull (IEI8_Type p_IEI := '50'O) :=

  {                         /* 24.501 cl. 9.11.3.44 */

    iei       := p_IEI, /* present in case of TV or TLV; omit in case of V */

    iel       := '02'O, /* present in case of LV or TLV; omit in case of V */

    psi0_7    := '00000000'B,                       /* PSI(0)- PSI(7) */

    psi8_15   := '00000000'B,                        /* PSI(8)- PSI(15)*/

    spare     := omit

  }
2.5 Change 5
	Function name
	f_TC_6_2_1_5_NR5GC_EUTRA

	Reason for change
	1. Because the NR registration process in the preamble step enables the UE to save the security context on the NR side, it should reset the security function of LTE side.

	Summary of change
	1. Add f_EUTRA_SS_NAS_SecurityRelease after LTE preamble step.

	TTCN module
	Idle_PLMNSelection_NR5GC_EUTRA.ttcn

	MCC160 Comment
	Accepted


Before change

function f_TC_6_2_1_5_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Inter-RAT Background HPLMN Search / Search for correct RAT for HPLMN / Automatic Mode

    f_EUTRA38_IRAT_Init(c1);

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell1, cs_HPLMN_001_11);//PLMN15

    // Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Preamble: The UE is in state Switched OFF (state 0N-B) according to TS 38.508-1

    f_EUTRA38_NR5GC_PreambleAndSwitchOff (eutra_Cell1);

    f_EUTRA_TestBody_Set(true);

    fl_TC_6_2_1_5_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    // take down the NR cell

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }
After change

function f_TC_6_2_1_5_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Inter-RAT Background HPLMN Search / Search for correct RAT for HPLMN / Automatic Mode

    f_EUTRA38_IRAT_Init(c1);

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell1, cs_HPLMN_001_11);//PLMN15

    // Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Preamble: The UE is in state Switched OFF (state 0N-B) according to TS 38.508-1

    f_EUTRA38_NR5GC_PreambleAndSwitchOff (eutra_Cell1);

    f_EUTRA_SS_NAS_SecurityRelease();
    f_EUTRA_TestBody_Set(true);

    fl_TC_6_2_1_5_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    // take down the NR cell

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }
2.6 Change 6
	Function name
	fl_TC_6_2_1_5_NR5GC_EUTRA_TestBody

	Reason for change
	1. HPLMN Timer should be started after UE completes registration.
2. The complete protocolConfigurationOptions information is not carried in the attach accept message
3. In order to match the change 2.2,change the information transfer step of step 21

	Summary of change
	1. Move f_EUTRA_NR_SendCoOrd after step 12b.
2. Replace f_EUTRA_Preamble with f_EUTRA38_NR5GC_Preamble in step 3.
3. Replace f_EUTRA38_NR5GC_S1ToN1_IdleMode with f_EUTRA38_Send_IRATParametersToNR in step 21.

	TTCN module
	Idle_PLMNSelection_NR5GC_EUTRA.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted

3. Accepted


Before change

function fl_TC_6_2_1_5_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    // The SS adjusts the NR and E-UTRAN Cell power levels according to row T0

    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE);

    // @siclog "Step 1" siclog@

    // Power on the UE

    f_UT_SwitchOnUE(UT);

    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_Trigger);
    // @siclog "Step 3-12b8" siclog@

    // Steps 2-10b8 of the registration procedure described in TS 38.508-1 table 4.5.2.2-1 take place on E-UTRAN Cell 1

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    // @siclog "Step 13-18 void" siclog@

    // @siclog "Step 21-27a1" siclog@

    // Generic test procedure in TS 38.508-1 Table 4.9.9.2.2-1 take place and UE is camped on NR Cell 1

    f_EUTRA38_NR5GC_S1ToN1_IdleMode(eutra_Cell1);
  }
After change

function fl_TC_6_2_1_5_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    // The SS adjusts the NR and E-UTRAN Cell power levels according to row T0

    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE);

    // @siclog "Step 1" siclog@

    // Power on the UE

    f_UT_SwitchOnUE(UT);

    // @siclog "Step 3-12b8" siclog@

    // Steps 2-10b8 of the registration procedure described in TS 38.508-1 table 4.5.2.2-1 take place on E-UTRAN Cell 1

    f_EUTRA38_NR5GC_Preamble(eutra_Cell1, TESTMode_OFF, RRC_IDLE);
    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_Trigger);
    // @siclog "Step 13-18 void" siclog@

    // @siclog "Step 21-27a1" siclog@

    // Generic test procedure in TS 38.508-1 Table 4.9.9.2.2-1 take place and UE is camped on NR Cell 1

    f_EUTRA38_Send_IRATParametersToNR(eutra_Cell1);
  }
3 Branches executed

4 Execution Log Files

4.1 HiSilicon Balong 5000
The HiSilicon Balong 5000 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610 and Starpoint SP9500 Test system. The documentation below is enclosed as evidence of the successful test case run [1]:
       Datang Linktester

· Test Case Execution log file:
\ TC_6_2_1_5_NR5GC_N41-FDD3.log
· PIXCIT settings used:
\ Pics(Hisilicon-IRAT).xml

\Pixit(TC_6_2_1_5_NR5GC).xml
Starpoint

· Test Case Execution log file:
\ TC_6_2_1_5_NR5GC_PASS.spm
· PIXCIT settings used:
\ NR5GC_IRAT_Testsuite.sppar

